Pakistan Journal of Hydrocarbon Research
Vol.2, No.1(January 1990), P.83-85

Automatic Computation of Schlumberger Apparent Resistivity Curves

Javed Akhter[1]

ABSTRACT

Geophysical prospecting by the apparent resistivity method involves the measurement of apparent resisitvity between two potential electrodes as a function of the separation of two collinear current electrodes. From the early days of the application of the resistivity sounding method, an important tool in the interpretation of the data has been the comparison of the apparent resistivity curves obtained from the field observations with apparent resistivity curves computed for assumed models of the layered earth consisting of two or three horizontal, homogeneous, and isotropic layers of given thickness and resistivity. Theoretical curves for models of more than two or three layers are approximated using published nomograms.

A modified version of the Koefoed (1979) program for the computation of the sample values of the apparent resistivity and current electrode spacing alongwith resistivity transform in a Schlumberger configuration, using the linear filter method by Gosh (1971) is presented. This program has been written in Turbo BASIC language and is based on the sampling interval of the fourth of a decade. By the twofold application of the filter, the sampling values of the apparent resistivity are computed at intervals of an eighth of a decade. In this trial and error method, the entire apparent resistivity curve of a model layered structure is calculated for comparison with the measured curve.

INTRODUCTION

The apparent resistivity $\rho a(r)$ for Schlumberger VES sounding in a horizontally layered earth is obtained as follows:

$$\rho a(r) = r^2 \int_0^\infty T(\lambda)\, J_1(\lambda r)\, \lambda d\lambda \quad \text{................(1)}$$

where r is half the current electrode spacing and $J_1(\lambda r)$ is a Bassel function of order 1. $T(\lambda)$ is known as the resistivity transform and is easily obtained for an arbitrary number of layers having different resistivities and thicknesses by means of recurrence formulas.

The linear filter method developed by Gosh (1971) permits the right hand side of equation (1) to be calculated using a set of fixed co-efficients taking advantage of the fact that $T(\lambda)$ in equation (1) is an algebraic expression involving no more complicated functions than the exponential.

In this algorithm, a point filter is used to calculate the apparent resistivity. The maximum error of calculation is less than 0.5% and in most practical cases it is much less. Unfortunately, numerical problems can arise if very large resistivity contrasts (of the order of 10^3 - 10^4 or larger) occur in two successive layers and the resistivity of the lower layer is the larger one. However, this problem is unlikely to arise in connection with the resistivities usually encountered in nature.

The electrical resistivity soundings are interpreted by matching them with a set of model resitivity curves. Each model curve corresponds with a system of subsurface layers of varying thickness and electrical resistivity. In practice, interpretation of resistivity soundings is invariably subjected to the principle of equivalence: any resistivity sounding can be matched with several slightly deviating model curves, representing different subsurface resistivity stratifications (Kunetz, 1966). Therefore, selection of a matching layer model should be subjected to constraints based on available geological information (Zohdy et al, 1974). Most useful in this respect are geophysical well logs, particularly the electric resistivity logs (Scot Keys and Mac Cary, 1971). Information may also be derived from lithologic logs, geologic maps, and data on subsurface water levels and electrical conductivities.

PROGRAM LISTING

This program computes apparent resistivity model curves and first asks for information concerning the layer model: the number of layers, their resistivities and their thicknesses. The parameters of the layers should be presented to the computer in the following sequence: first the top layer, and then going down step by step.

The program then asks for information concerning the output you want: the abscissa of first point of output and the number of samples of the output (at intervals of an eighth of a decade). The values of the apparent resistivity at these sample points are then printed out.

[1] Geological Survey of Pakistan, Islamabad

```
5 REM Schlumberger Apparent Resistivity Model
  Curves
10 DIM R(10), D(9), H(9), T(35)
20 F = EXP(LOG (10)/8)
30 INPUT "GIVE NUMBER OF LAYERS=";I9
40 I8 = I9 - 1
60 PRINT "GIVE RESISTIVITIES"
80 FOR I = 1 TO I9
90 INPUT R(I)
100 NEXT I
110 PRINT "GIVE DEPTHS"
130 FOR I = 1 TO I8
140 INPUT H(I)
150 NEXT I
152 D(1) =H(1)
153 FOR I =1 TO I8
154 D(I + 1) = H(I + 1) -H(I)
155 NEXT I
160 PRINT "LAYER RESISTIVITY DEPTH"
180 FOR I = 1 TO I8
190 PRINT TAB (2);I;TAB(10);R(I);TAB(30);H(I)
210 NEXT I
220 PRINT TAB(2);I9; TAB(15);R(I9)
240 INPUT "GIVE FIRST ABSCISSA =";X1
260 PRINT "FIRST ABSCISSA=";X1
280 INPUT "GIVE NUMBER OF SAMPLES=";N
300 PRINT "NUMBER OF SAMPLES=";N
330 PRINT "RESISTIVITY TRANSFORM
  FUNCTION"
335 PRINT "NO  RAMDA        T(J) "
340 Y = X1/822.8
350 FOR J = 1 TO 34
360 GOSUB 550
370 T(J) = B
380 Y = Y *F
385 PRINT TAB (2);J; TAB (10); Y; TAB(35); T(J)
390 NEXT J
392 PRINT "Schlumberger VES Curve"
395 PRINT " NO        L/2         AR
400 FOR M = 1 TO N
410 GOSUB 550
420 T(35) = B
430 Y = Y* F
440 S=42*T(1)-103*T(3)+144*T(5)-211*T(7)+
  330*T(9)-574*T(11)
450 S=S+1184*T(13)-3162*T(15)+10219*T(17)-
  24514*T(19)
460 S=S+18192*T(21)+6486*T(23)+1739*T(25)
  +79*T(27)+200*T(29)
470 S=(S-106*T(31)+93*T(33)-38*T(35))/10000
480 FOR J =1 to 34
490 T(J) = T(J +1)
500 NEXT J
510 AO = Y/28.8207792
520 PRINT TAB(2);M; TAB(20);AO;TAB(47);S
530 NEXT M
540 STOP
550 B = R(I9)
560 FOR K = 1 TO I8
570 I = I9 - K
580 U = D(I)/Y
590 IF (5 -U) > O THEN 620
600 B = R(I)
610 GOTO 650
620 A1 = EXP (U)
630 A2 = (A1-1/A1)/(A1+1/A1)
640 B = (B+A2*R(I))/(1 +A2*B/R(I))
650 NEXT K
660 RETURN
670 END
```

TEST EXAMPLE

```
GIVE NUMBER OF LAYERS = 3
GIVE RESISTIVITIES
? 50
? 350
? 100
GIVE DEPTHS
? 1
? 4
GIVE FIRST ABSCISSA = ? 1
FIRST ABSCISSA = 1
GIVE NUMBER OF SAMPLES = ? 30
NUMBER OF SAMPLES = 30
```

Resistivity No.	Transform RAMDA	Functions T (J)
1	1.62E-03	50
2	2.16E-03	50
3	2.88E-03	50
4	3.84E-03	50
5	5.12E-03	50
6	6.83E-03	50
7	9.11E-03	50
8	0.012	50
9	0.016	50
10	0.021	50
11	0.028	50
12	0.038	50
13	0.051	50
14	0.068	50
15	0.091	50
16	0.012	50
17	0.162	50
18	0.216	50
19	0.288	50
20	0.384	50.072
21	0.512	50.413

22	0.683	51.537
23	0.911	54.187
24	1.215	59.114
25	1.620	66.868
26	2.161	77.644
27	2.882	91.068
28	3.843	105.940
29	5.125	120.211
30	6.834	131.512
31	9.113	138.158
32	12.153	139.859
33	16.207	137.770
34	21.612	133.360

Schlumberger VES Curve

No	L/2	AR
1	0.999	57.698
2	1.333	66.956
3	1.778	76.547
4	2.371	92.534
5	3.162	111.714
6	4.216	131.935
7	5.623	150.264
8	7.498	163.106
9	10.000	167.112
10	13.335	161.053
11	17.782	147,332
12	23.713	131.256
13	31.622	117.910
14	42.169	109.375
15	56.234	104.824
16	74.989	102.550
17	100.00	101.380
18	133.352	100.762
19	177.827	100.430
20	237.137	100.240
21	316.227	100.133
22	421.696	100.073
23	562.341	100.035
24	749.894	100.005
25	1000.000	99.981
26	1333.521	99.964
27	1778.279	99.950
28	2371.373	99.940
29	3162.277	99.942
30	4216.965	99.956

CONCLUSIONS

A BASIC computer program is presented for comparison of resistivity field data with the data derived from a layer model. If the agreement between the two sets of data is unsatisfactory, then the parameters of the layer model are adjusted. The processing is iterated until a sufficient agreement between these model data and field data is obtained. A version of this method, in which the adjustment of the layer parameters is done by human judgment, has been a common practice in resisitivity sounding interpretation for many years.

REFERENCES

Gosh, D.P., 1971, Inverse filter coefficients for the computation of apparent resistivity standard curves for a horizontally stratified earth: Geophysical Prospecting, v.19, p.765-775.

Koefoed, O., 1979: Geosounding principles, Part 1, Elsevier Publishing Company, Amsterdam, 276p.

Kunetz, G., 1966: Principles of direct current resistivity prospecting, Berlin, Gebruder Borntraeger, 103p.

Scott Keys, W., and L.M. Mac Cary, 1971: Applications of borehole geophysics to water resources investigations, United States Government Printing Office, Washington, 126p.

Zohdy, A.A.R., G.P. Eaton and D.R. Mabey, 1974: Application of surface geophysics to groundwater investigations, United States Government Printing Office, Washington, 116p.